
Subject: Re: IDLgrSurface with transparent lines?
Posted by [Karl Schultz](#) on Thu, 29 Apr 2004 14:05:55 GMT
[View Forum Message](#) <> [Reply to Message](#)

"Antonio Santiago" <d6522117@est.fib.upc.es> wrote in message
news:4090B32E.1040409@est.fib.upc.es...

> Hi,

>

> is there a way i can draw an IDLgrSurface object with STYLE=1 (line
> style) and this lines be semi transparen, like a transparent texture
image.

A sort of quiet feature in IDL 6.0 is that texture mapping is applied to
lines. The following test demonstrates this ability:

pro test

```
img = BYTARR(4,2,2)
img[1,*,*] = 255 ; green
img[3,*,*] = 100 ; alpha
oTexture = OBJ_NEW('IDLgrImage', img)
oSurface = OBJ_NEW('IDLgrSurface', DIST(45), $
    COLOR=[255,255,255], STYLE=1, THICK=3, TEXTURE_MAP=oTexture)
READ_JPEG, FILEPATH('rose.jpg', SUBDIR=['examples','data']), img, /TRUE
oTexture2 = OBJ_NEW('IDLgrImage', img)
oPolygon = OBJ_NEW('IDLgrPolygon', [0,1,1,0]*45, [0,0,1,1]*45, $
    COLOR=[255,255,255], TEXTURE_MAP=oTexture2, $
    TEXTURE_COORD=[[0,0],[1,0],[1,1],[0,1]])
oModel = OBJ_NEW('IDLgrModel')
oModel->Add, [oPolygon, oSurface]
XOBJVIEW, oModel
```

end

If you leave the XOBJVIEW settings at their default values, then when you
click on the model to rotate it, the "drag quality" goes to medium which
lets you easily see what it looks like without transparency.

There will be an easier way to do this in IDL 6.1.

Karl
